LU TP 78-18
November, 1978

A Monte Carlo Program for Quark Jet Generation

T. Sjöstrand, B. Söderberg

Department of Theoretical Physics
University of Lund
Lund, Sweden

Abstract:

A Monte Carlo computer program is presented, that simulates the fragmentation of a fast quark into a jet of mesons. It uses an iterative scaling scheme, and is compatible with the jet model of Field and Feynman.

1. Introduction

There seems to be a growing belief in jets as a useful concept in deep inelastic scattering. Furthermore, the jet structure appears to be basically the same whether the jets come from hadron-hadron collisions or are produced in collisions involving leptons. The jets are believed to arise from the fragmentation of quarks (or gluons) into hadrons, going in approximately the same direction as the original parton.

To date, quantum chromodynamics is not developed enough to give a detailed description of jets. This has prompted the construction of phenomenological models, by Field and Feynman [1] and by Andersson, Gustafson and Peterson [2], among others. The model used in our program is compatible with both these models.

The basic idea is that, when a fast quark emerges, a quark-antiquark pair will be created in the colour field in its wake. The antiquark together with the initial quark form a meson, while the left-over quark provides the starting point for the rest of the jet. Unstable primary mesons decay and, finally, particles going backwards are excluded in order to make the jet finite.

Note that the model only describes the fragmentation of a quark; it says nothing about how to get the quark.

In chapter 2 we discuss the model used in the generation of the primary mesons, and in chapter 3 the scheme chosen for the subsequent decay of unstable mesons.

Chapter 4 contains a closer look at the different FORTRAN program components. In chapter 5 we give some general information on how to use the package.

Complete computer listings of the program components are included in the Appendix.

2. The model

A quark of a given flavour 'a' comes out from a reaction. Out of the colour field materializes a quark-antiquark pair '$b\bar{b}$' . The quark 'a' and the antiquark '$\bar{b}$' form a primary meson '$a\bar{b}$' . From the field of the left-over quark 'b' another pair '$c\bar{c}$' is created, leading to a second primary meson '$b\bar{c}$' being formed. Iterating this process gives a chain of primary mesons '$a\bar{b}$' , '$b\bar{c}$' , '$c\bar{d}$' , etc.

With a spin 0 or 1 assigned to each meson, its type is defined, except for the cases $u\bar{u}$, $d\bar{d}$ and $s\bar{s}$, where mixing has to be taken into account.

Every quark or antiquark is supposed to have a transverse (relative to the jet axis) momentum $\bar{p}_\perp$, as in [1] randomly distributed according to

$$f_q(\bar{p}_\perp)\, d^2p_\perp = \frac{1}{\pi\sigma^2}\, e^{-(p_\perp^2/\sigma^2)}\, d^2p_\perp$$

with the constraint that the total $\bar{p}_\perp$ of any quark-antiquark pair ($b\bar{b}$, $c\bar{c}$, ...) be zero. This leads to a $\bar{p}_\perp$ distribution for the primary mesons which also is a Gaussian

$$f_m(\bar{p}_\perp)\, d^2p_\perp = \frac{1}{2\pi\sigma^2}\, e^{-(p_\perp^2/2\sigma^2)}\, d^2p_\perp$$

It also implies a $\bar{p}_\perp$ correlation between consecutive primary mesons.

As long as the energy of the system is very large, the choice of scaling variable is not crucial. However, when the jet energy comes down to the order of a meson mass, the difference becomes important. In the present model, $W = E + p_z$ ($c=1$, jet along the z axis) is preferred to p_z, because scaling in W is invariant under a Lorentz boost in the z direction, in contrast to the case for p_z. Furthermore, $\frac{dp_z}{E} = \frac{dW}{W}$ is a natural phase-space element, giving a reasonable plateau behaviour.

So, assuming the initial quark to have $E + p_z$ equal to W_o, the first primary meson takes a fraction $x_1 W_o$, leaving $W_1 = (1 - x_1)W_o$ to the remaining quark. The second meson in its turn takes $x_2 W_1$, and so on. The x is in every step randomly distributed between 0 and 1 according to

$$f_x(x)\,dx = (1 - a + 3a(1-x)^2)\,dx$$

which, with different values of a, gives the distributions of [1] and [2].

The variables W and $\bar{p}_\perp$ for a meson thus being defined, E and p_z are obtained from

$$E + p_z = W$$

$$E - p_z = \frac{m_\perp^2}{W}$$

where $m_\perp = \sqrt{p_\perp^2 + m^2}$ is the transverse mass of the meson.

Of course, a large part of the primary mesons thus produced will be unstable and subsequently decay into stable final particles.

The scheme above produces an infinite number of particles, most of them having a small W and thus a large negative p_z. This problem is resolved by keeping only final particles with a non-negative p_z, which gives a physical jet that can be compared with experiments. In effect, this jet will have a finite mean multiplicity $\propto \log W$.

Energy, momentum and flavour are not exactly conserved quantities in this scheme. However, a jet is only one part of the final state of a reaction, and does not by itself have to conserve these quantities.

3. The decay scheme

Decay channels and branching ratios are taken from [3].

For a two-particle decay, a spherically symmetric angular distribution, in the rest frame of the decaying particle, is used.

A three-particle decay is formulated in terms of a double two-particle decay

$$m \rightarrow m_1 + m_x$$
$$\hookrightarrow m_2 + m_3$$

and the final state distribution is expressed in the distribution of m_x .

In most cases, pure phase-space is used, leading to an m_x probability distribution

$$g_o(m_x^2)\,dm_x^2 \propto \frac{1}{m_x^2}\sqrt{\lambda(m^2,m_1^2,m_x^2)\,\lambda(m_x^2,m_2^2,m_3^2)}\,dm_x^2$$

(where $\lambda(a,b,c) = a^2 + b^2 + c^2 - 2ab - 2ac - 2bc$)
and both two-particle decays being spherically symmetric in the respective rest frame.

However, in the two cases

$$\omega \rightarrow \pi^+\pi^-\pi^o$$
$$\phi \rightarrow \pi^+\pi^-\pi^o$$

we take more care in order to get a reasonable angular distribution for the π mesons. Since ω and ϕ have isospin 0 and the π mesons isospin 1 , the isospin part of the matrix element is antisymmetric under an interchange of any two π's in the final state. Thus, the spatial part must also be antisymmetric in order to make the total matrix element symmetric, as it should be for bosons.

Furthermore, from parity considerations and the fact that ω and ϕ are vector mesons follows that it must transform as an axial vector. The simplest choice for the spatial matrix element then is [4]

$$\langle 3\pi | S | \omega \rangle \propto \bar{p}_1 \times \bar{p}_2 + \bar{p}_2 \times \bar{p}_3 + \bar{p}_3 \times \bar{p}_1$$

where the $\bar{p}_i$ of the pions are to be evaluated in the rest frame of the ω or ϕ. Squaring, and averaging over spin components, we get

$$|\bar{p}_1 \times \bar{p}_2|^2 \propto \frac{1}{m_x^2} \lambda(m^2, m_1^2, m_x^2) \lambda(m_x^2, m_2^2, m_3^2) \sin^2\alpha$$

where α is the angle between $\bar{p}_1$ and $\bar{p}_2$ in the rest frame of m_x. Inserting phase-space factors gives

$$g_1(m_x^2)\, dm_x^2 \propto \frac{1}{m_x^4} \left\{ \lambda(m^2, m_1^2, m_x^2) \lambda(m_x^2, m_2^2, m_3^2) \right\}^{3/2} dm_x^2$$

and an angular probability distribution

$$h(\cos\alpha)\, d(\cos\alpha) \propto (1 - \cos^2\alpha)\, d(\cos\alpha)$$

in the m_x decay.

4. The program components

The program package consists of five parts written in FORTRAN, the first four of which are subroutines and the last one a block data.

1) JETGEN, which generates the primary mesons of the jet,
2) DECAY, which handles the decay into stable particles, called by JETGEN,

3) EDIT, which may be used to throw away uninteresting particles or make overall rotations or Lorentz boosts,

4) LIST, which lists the particles of the jet,

5) BLOCK DATA, containing tables and default values.

Of major importance is the commonblock JET, where an event is stored. Each filled row in the K and P matrices corresponds to one particle. For the I:th particle produced, K(I,1) describes its origin, K(I,2) gives the particle type and P(I,1) through P(I,5) give p_x , p_y , p_z , E and m , respectively (Table 1). The argument N , in the JETGEN, EDIT and LIST calls, signifies that the N first rows are filled with valid information.

The particles γ , $\pi^{\pm}$, $K^{\pm}$, K^{0} and $\bar{K}^{0}$ are, for our purposes, called stable, since they do not decay through strong or electromagnetic interactions. In the program we put c=1 and give energies, momenta and masses in MeV.

RANF is a random number generator, with 0 < RANF(0) < 1 .

4a JETGEN

With a CALL JETGEN(N) we produce a jet in the z direction, given the initial quark flavour IFLBEG (Table 4) and energy EBEG . A negative IFLBEG signifies an antiquark jet, which is produced exactly like the quark one, except in step 3 below. IFLBEG , EBEG and the parameters of the model are collected in the commonblock PAR and given default values in BLOCK DATA (Table 2).

Before beginning the jet generation, we calculate IFLSGN , used in step 3, and the initial value of W , the $E + p_z$ available. By starting with $W = 2 \cdot EBEG$, it turns out that the physical jet, consisting of the stable particles with $p_z \geq 0$, will have a total energy in the order of EBEG . Also, I , counting the number of particles produced, and IDP , counting the ones having been checked for stability, are put equal to zero.

Step 1: The flavour for the first quark is defined. The quark is also given a transverse momentum using the probability distribution $f_q(\tilde{p}_\perp)$, with σ or SIGMA as a free parameter.

Step 2: A quark-antiquark pair is materialized out of the colour field with probability PUD for the pair being $u\bar{u}$, the same for $d\bar{d}$ and 1-2•PUD for $s\bar{s}$. The transverse momentum for the antiquark is chosen, again using $f_q(\tilde{p}_\perp)$.

Step 3: Combining the "old" quark with the "new" antiquark, we form a meson flavour, which, after renumbering provided by MESO with IFLSGN=1 , is stored in K(I,1) (Table 5). For an antiquark jet, IFLSGN=2 , the renumbering is done differently to compensate for the change of rôles between quarks and antiquarks. The meson is assigned spin, with probability PS1 for spin 1 and 1-PS1 for spin 0 . This determines the particle type K(I,2) (Table 6), except for the meson flavours $u\bar{u}$, $d\bar{d}$ and $s\bar{s}$; the parameters of mixing (Table 5) are stored in CMIX as follows. For the spin and flavour combination

KM , CMIX(KM,1) is the probability to get either η or η' for spin 0, either ω or ϕ for spin 1, while CMIX(KM,2) is the probability for η' , ϕ alone.

Step 4: The meson mass is taken from PMAS (Table 6) and stored in P(I,5) . Note that we use sharp mass values. The transverse momentum is taken as the vectorial sum of the quark transverse momenta, which gives the distribution $f_m(\bar{p}_\perp)$, with $\langle p_\perp \rangle = \sqrt{\frac{\pi}{2}}\,\sigma$ and $\langle p_\perp^2 \rangle = 2\sigma^2$. The resulting p_x and p_y are stored in P(I,1) and P(I,2) , respectively, and the transverse mass squared $m_\perp^2$ is calculated.

Step 5: We choose x according to the probability distribution $f_x(x)$, with CX2 as the free parameter a . Using $m_\perp^2$, we separate out p_z and E , which are stored in P(I,3) and P(I,4) .

Step 6: If the meson is unstable, we call the subroutine DECAY , which handles the decay into secondaries. These are then allowed to decay into tertiaries, and so on, until all decay chains have been traced down to stable particles, or we don't have any space left for more particles.

Step 7: The quark appearing in step 2 is now at the end of the string. We assume that the quark-antiquark pair was created with total flavour and transverse momentum zero, giving the values that will be needed in the next round of the generation.

Step 8: The remaining W is calculated, and if W is larger than WFIN and if we have space left, we go back to step 2.

Finally, before returning to the main program, we put N equal to the number of rows entered, i.e. particles created.

4b DECAY

A CALL DECAY(IPD,I) in JETGEN lets the particle in row IPD decay, without disturbing this row, and stores the products immediately below row I , the latter variable being updated in the end. This is done in four steps.

Step 1:

Our table of decay channels and branching ratios (Table 7) is stored in KDP and CBR . For a given decay channel number IDC , the decay products are given by KDP(IDC,J) , J=1,2,3 , where KDP(IDC,3) = 0 in a two-particle decay. CBR(IDC) contains the sum of branching ratios up to and including channel IDC . The method to choose decay channel is to fix a random number, enter the table at the proper point, given by IDCO for each unstable particle, and go down the table until CBR becomes larger than the random number. Then we read out ND , the number of decay products and start the filling of the product rows. As with particles in JETGEN , K(I1,2) denotes type and P(I1,5) mass, but in K(I1,1) we will now store the row number of the decaying particle, with a minus sign to distinguish it from the use of this entry in JETGEN. This gives us a means of keeping track of all decay chains.

Step 2:

In order to treat a three-particle decay as two consecutive two-particle decays we need to find m_x . With $SX = m_x^2$ and SA , SB , SC , and SD as defined in the program, TDF gives the probability distribution in SX , either $g_0(SX)$ or $g_1(SX)$.

TDU provides an upper bound for TDF , and is used to choose SX by hit-or-miss Monte Carlo. The resulting value for m_x is stored in row 100 , which is used only for these nonphysical intermediate states.

Step 3:

Two-particle decays are now carried out in the respective center of mass frames, with the decaying particle in row I0 and the products in I1 and I2 . For a real two-particle decay this means, with indices referring to row numbers, $m_{IPD} \rightarrow m_{I+1} + m_{I+2}$, while for a three-particle one we first let $m_{IPD} \rightarrow m_{I+1} + m_{100}$ and then $m_{100} \rightarrow m_{I+2} + m_{I+3}$. In each case PA , the momentum of the decay products, is calculated, and a direction vector U is chosen on the unit sphere with probability distribution $\frac{1}{4\pi}\, d\Omega$ or, for the second half of the ω and ϕ 3π-decays, $\frac{3}{8\pi} \sin^2\alpha\, d\Omega$, where α is the angle between $\bar{p}_{I+1}$ and $\bar{p}_{I+2}$ in the rest frame of m_x . Finally, $\bar{p}$ and E for the decay products are calculated and stored.

Step 4:

Lorentz boosts are made as follows. For a three-particle decay we use the contents of row 100 to take the products 2 and 3

to the decaying meson rest frame. With all products in that rest frame, for two-particle as well as three-particle decays, we use row IPD to take them to the original frame.

Before returning to JETGEN , the value of I is also updated with the number of decay products, i.e. new rows added.

4c EDIT

A CALL EDIT(N) allows the jet stored in the commonblock JET to be altered in a number of ways by using the parameters in the commonblock EDPAR (Table 3).

Step 1:

Many of the particles in JET are either unstable or neutral, and may not be interesting when comparing with experiments. Depending on the value of ITHROW , different actions are taken. Default is to keep all the stable particles γ , $\pi^{\pm}$, $K^{\pm}$, K^{0} , and $\bar{K}^{0}$.

Independently of this, particles with p_z < PZMIN or p < PMIN are thrown away, in the first case to give a physical jet, using the default PZMIN = 0 , in the second one to help simulate detector effects. Data on the particles we want to keep are packed into the top rows. For decay products, K(I,1) will not keep its original meaning, so in order to avoid misunderstandings we put K(I,1) = 0 for them. In the end, the number of particles left replaces the old value of N .

Step 2:

The jet is produced in the z ,i.e. $\theta = 0$, direction by JETGEN , and is by default kept that way. For nonzero THETA(θ) and PHI(φ) , however, the corresponding rotation matrix is calculated and the $\bar{p}$:s of all particles are rotated to correspond to a jet produced in the θ,φ direction.

Step 3:

When producing a physical jet, this is usually done in some center of mass frame of the collision giving rise to the jet, which need not be the same as the lab frame. Therefore a possibility to make an overall Lorentz boost is also included. If the velocity of the CM relative to the lab frame is given by the vector BETA , $\bar{\beta} = \frac{\bar{v}}{c}$, this defines the boost that takes the jet to the lab frame.

4d LIST

The jet produced can be listed with a CALL LIST(N) .

The total information available in each of the N first rows of the commonblock JET is listed (Table 1). Column I gives the row number. ORI stands for the origin of the particle, as stored in K(I,1) . If K(I,1) > 0 , the original meson flavour is written (Table 5), while if K(I,1) $\leq$ 0 , we print -K(I,1) . PART gives particle type (Table 6) and STAB is written for the stable particles. Then comes p_x , p_y , p_z , E , and m , respectively.

4e BLOCK DATA

In BLOCK DATA we have collected default values and particle data.

The commonblocks PAR , used in JETGEN , and EDPAR , used in EDIT , have been assigned default values consistent with the Field-Feynman model (Table 2, Table 3).

The particle data in DATA1 and DATA2 have been selected from Review of Particle Properties [3]. Our simple numbers for the flavour mixing correspond to octet-singlet mixing angles $\theta_{pseudoscalar} = -9.7^{o}$ and $\theta_{vector} = 35.3^{o}$, values which deviate only slightly from the experimental fits. Decay channels with branching ratios below 1.2 % have not been included in DATA2 (Table 7), and the CBR values are slightly rescaled accordingly. The branching ratios for $\rho \rightarrow \pi\pi$ and $K^{*} \rightarrow K\pi$ have been calculated from isospin invariance.

DATA3 contains alphanumerical strings used in LIST.

5. How to use the program.

A jet is generated with a

```
CALL JETGEN(N)
```

where N is the number of particles produced and stored in the commonblock JET . After the call above, this will include both the unstable particles and their decay products, as well as particles with $p_z < 0$. With the sequence

```
CALL JETGEN(N)
CALL EDIT(N)
```

only the stable particles with $p_z \geq 0$ will survive, giving the physical jet, and N is changed accordingly by EDIT .

The jet may be listed by including

```
CALL LIST(N) .
```

For a thorough analysis of the jet, a card

```
COMMON/JET/ K(100,2) , P(100,5)
```

is needed among the declarations. The jet produced is then available in the N first rows of the K and P matrices (Table 1)

The subroutines are provided with sensible default values to facilitate their use. Thus, JETGEN together with EDIT will, by default, produce a Field-Feynman u jet with total energy approximately 10 GeV. This may be changed by including the declaration

```
COMMON /PAR/ PUD,PS1,SIGMA,CX2,EBEG,WFIN,IFLBEG
```

and giving new values where desired (Table 2). In particular, IFLBEG determines the flavour of the first quark and EBEG the approximate total energy of the physical jet. The model proposed by Andersson, Gustafson and Peterson is characterized by PS1 = 0.75 and CX2 = 0 .

EDIT is intended to extract a physical jet from the one produced by JETGEN. Its full capabilities are accessible by including a card

```
COMMON /EDPAR/ ITHROW,PZMIN,PMIN,THETA,PHI,BETA(3)
```

and changing the parameter values (Table 3). One application might be to produce a physical jet of given energy in the center

of mass frame, with some probability distribution for the direction of the jet axis, and then transform to the lab frame.

In principle, several EDIT calls can be made for each jet, but caution is then urged. For instance, the cut in p_z may have to be avoided, by giving PZMIN a large negative value, if the jet axis has been rotated in a previous call.

Finally, we note that other kinds of $p_\perp$ or x distributions are easy to accommodate with only minor modifications in JETGEN .

References

1. R.D. Field and R.P. Feynman,
Nucl. Phys. B136 (1978) 1

2. B. Andersson, G. Gustafson, C. Peterson,
A Semiclassical Model for Quark Jet Fragmentation
Lund preprint LU TP 78-5,
to be published in Z. Physik C Particles and Fields

3. Particle Data Group,
Phys. Lett. 75B (1978) 1

4. G. Källén,
Elementary Particle Physics
(Addison-Wesley, Reading, 1964).

Appendix

Listings of the program components.

```
      SUBROUTINE JETGEN(N)
      COMMON /JET/ K(100,2), P(100,5)
      COMMON /PAR/ PUD, PS1, SIGMA, CX2, EBEG, WFIN, IFLBEG
      COMMON /DATA1/ MESO(9,2), CMIX(6,2), PMAS(19)
      IFLSGN=(10-IFLBEG)/5
      W=2.*EBEG
      I=0
      IPD=0
C 1 FLAVOUR AND PT FOR FIRST @UARK
      IFL1=IABS(IFLBEG)
      PT1=SIGMA*SQRT(-ALOG(RANF(0)))
      PHI1=6.2832*RANF(0)
      PX1=PT1*COS(PHI1)
      PY1=PT1*SIN(PHI1)
  100 I=I+1
C 2 FLAVOUR AND PT FOR NEXT ANTI@UARK
      IFL2=1+INT(RANF(0)/PUD)
      PT2=SIGMA*SQRT(-ALOG(RANF(0)))
      PHI2=6.2832*RANF(0)
      PX2=PT2*COS(PHI2)
      PY2=PT2*SIN(PHI2)
C 3 MESON FORMED, SPIN ADDED AND FLAVOUR MIXED
      K(I,1)=MESO(3*(IFL1-1)+IFL2,IFLSGN)
      ISPIN=INT(PS1+RANF(0))
      K(I,2)=1+9*ISPIN+K(I,1)
      IF(K(I,1).LE.6) GOTO 110
      TMIX=RANF(0)
      KM=K(I,1)-6+3*ISPIN
      K(I,2)=8+9*ISPIN+INT(TMIX+CMIX(KM,1))+INT(TMIX+CMIX(KM,2))
C 4 MESON MASS FROM TABLE, PT FROM CONSTITUENTS
  110 P(I,5)=PMAS(K(I,2))
      P(I,1)=PX1+PX2
      P(I,2)=PY1+PY2
      PMTS=P(I,1)**2+P(I,2)**2+P(I,5)**2
C 5 RANDOM CHOICE OF X=(E+PZ)MESON/(E+PZ)AVAILABLE GIVES E AND PZ
      X=RANF(0)
      IF(RANF(0).LT.CX2) X=1.-X**(1./3.)
      P(I,3)=(X*W-PMTS/(X*W))/2.
      P(I,4)=(X*W+PMTS/(X*W))/2.
C 6 IF UNSTABLE, DECAY CHAIN INTO STABLE PARTICLES
  120 IPD=IPD+1
      IF(K(IPD,2).GE.8) CALL DECAY(IPD,I)
      IF(IPD.LT.I.AND.I.LE.96) GOTO 120
C 7 FLAVOUR AND PT OF @UARK FORMED IN PAIR WITH ANTI@UARK ABOVE
      IFL1=IFL2
      PX1=-PX2
      PY1=-PY2
C 8 IF ENOUGH E+PZ LEFT, GO TO 2
      W=(1.-X)*W
      IF(W.GT.WFIN.AND.I.LE.95) GOTO 100
      N=I
      RETURN
      END
```

```
      SUBROUTINE DECAY(IPD,I)
      COMMON /JET/ K(100,2), P(100,5)
      COMMON /DATA1/ MESO(9,2), CMIX(6,2), PMAS(19)
      COMMON /DATA2/ IDCO(12), CBR(29), KDP(29,3)
      DIMENSION U(3), BE(3)
C 1 DECAY CHANNEL CHOICE, GIVES DECAY PRODUCTS
      TBR=RANF(0)
      IDC=IDCO(K(IPD,2)-7)
  100 IDC=IDC+1
      IF(TBR.GT.CBR(IDC)) GOTO 100
      ND=(59+KDP(IDC,3))/20
      DO 110 I1=I+1,I+ND
      K(I1,1)=-IPD
      K(I1,2)=KDP(IDC,I1-I)
  110 P(I1,5)=PMAS(K(I1,2))
C 2 IN THREE-PARTICLE DECAY CHOICE OF INVARIANT MASS OF PRODUCTS 2+3
      IF(ND.EQ.2) GOTO 130
      SA=(P(IPD,5)+P(I+1,5))**2
      SB=(P(IPD,5)-P(I+1,5))**2
      SC=(P(I+2,5)+P(I+3,5))**2
      SD=(P(I+2,5)-P(I+3,5))**2
      TDU=(SA-SD)*(SB-SC)/(4.*SQRT(SB*SC))
      IF(K(IPD,2).GE.11) TDU=SQRT(SB*SC)*TDU**3
  120 SX=SC+(SB-SC)*RANF(0)
      TDF=SQRT((SX-SA)*(SX-SB)*(SX-SC)*(SX-SD))/SX
      IF(K(IPD,2).GE.11) TDF=SX*TDF**3
      IF(RANF(0)*TDU.GT.TDF) GOTO 120
      P(100,5)=SQRT(SX)
C 3 TWO-PARTICLE DECAY IN CM, TWICE TO SIMULATE THREE-PARTICLE DECAY
  130 DO 160 IL=1,ND-1
      IO=(IL-1)*100-(IL-2)*IPD
      I1=I+IL
      I2=(ND-IL-1)*100-(ND-IL-2)*(I+IL+1)
      PA=SQRT((P(IO,5)**2-(P(I1,5)+P(I2,5))**2)*
     &(P(IO,5)**2-(P(I1,5)-P(I2,5))**2))/(2.*P(IO,5))
  140 U(3)=2.*RANF(0)-1.
      PHI=6.2832*RANF(0)
      U(1)=SQRT(1.-U(3)**2)*COS(PHI)
      U(2)=SQRT(1.-U(3)**2)*SIN(PHI)
      TDA=1.-(U(1)*P(IO,1)+U(2)*P(IO,2)+U(3)*P(IO,3))**2/
     &(P(IO,1)**2+P(IO,2)**2+P(IO,3)**2)
      IF(K(IPD,2).GE.11.AND.IL.EQ.2.AND.RANF(0).GT.TDA) GOTO 140
      DO 150 J=1,3
      P(I1,J)=PA*U(J)
  150 P(I2,J)=-PA*U(J)
      P(I1,4)=SQRT(PA**2+P(I1,5)**2)
  160 P(I2,4)=SQRT(PA**2+P(I2,5)**2)
C 4 DECAY PRODUCTS LORENTZ TRANSFORMED TO LAB SYSTEM
      DO 190 IL=ND-1,1,-1
      IO=(IL-1)*100-(IL-2)*IPD
      DO 170 J=1,3
  170 BE(J)=P(IO,J)/P(IO,4)
      GA=P(IO,4)/P(IO,5)
      DO 190 I1=I+IL,I+ND
      BEP=BE(1)*P(I1,1)+BE(2)*P(I1,2)+BE(3)*P(I1,3)
      DO 180 J=1,3
  180 P(I1,J)=P(I1,J)+GA*(GA/(1.+GA)*BEP+P(I1,4))*BE(J)
  190 P(I1,4)=GA*(P(I1,4)+BEP)
      I=I+ND
      RETURN
      END
```

```
      SUBROUTINE EDIT(N)
      COMMON /JET/ K(100,2), P(100,5)
      COMMON /EDPAR/ ITHROW, PZMIN, PMIN, THETA, PHI, BETA(3)
      REAL ROT(3,3), PR(3)
C 1 THROW AWAY NEUTRALS OR UNSTABLE OR WITH TOO LOW PZ OR P
      I1=0
      DO 110 I=1,N
      IF(ITHROW.GE.1.AND.K(I,2).GE.8) GOTO 110
      IF(ITHROW.GE.2.AND.K(I,2).GE.6) GOTO 110
      IF(ITHROW.GE.3.AND.K(I,2).EQ.1) GOTO 110
      IF(P(I,3).LT.PZMIN.OR.P(I,4)**2-P(I,5)**2.LT.PMIN**2) GOTO 110
      I1=I1+1
      K(I1,1)=IDIM(K(I,1),0)
      K(I1,2)=K(I,2)
      DO 100 J=1,5
  100 P(I1,J)=P(I,J)
  110 CONTINUE
      N=I1
C 2 ROTATE TO GIVE JET PRODUCED IN DIRECTION THETA, PHI
      IF(THETA.LT.1E-4) GOTO 140
      ROT(1,1)=COS(THETA)*COS(PHI)
      ROT(1,2)=-SIN(PHI)
      ROT(1,3)=SIN(THETA)*COS(PHI)
      ROT(2,1)=COS(THETA)*SIN(PHI)
      ROT(2,2)=COS(PHI)
      ROT(2,3)=SIN(THETA)*SIN(PHI)
      ROT(3,1)=-SIN(THETA)
      ROT(3,2)=0.
      ROT(3,3)=COS(THETA)
      DO 130 I=1,N
      DO 120 J=1,3
  120 PR(J)=P(I,J)
      DO 130 J=1,3
  130 P(I,J)=ROT(J,1)*PR(1)+ROT(J,2)*PR(2)+ROT(J,3)*PR(3)
C 3 OVERALL LORENTZ BOOST GIVEN BY BETA VECTOR
  140 IF(BETA(1)**2+BETA(2)**2+BETA(3)**2.LT.1E-8) RETURN
      GA=1./SQRT(1.-BETA(1)**2-BETA(2)**2-BETA(3)**2)
      DO 160 I=1,N
      BEP=BETA(1)*P(I,1)+BETA(2)*P(I,2)+BETA(3)*P(I,3)
      DO 150 J=1,3
  150 P(I,J)=P(I,J)+GA*(GA/(1.+GA)*BEP+P(I,4))*BETA(J)
  160 P(I,4)=GA*(P(I,4)+BEP)
      RETURN
      END
```

```
      SUBROUTINE LIST(N)
      COMMON /JET/ K(100,2), P(100,5)
      COMMON /DATA3/ CHA1(9), CHA2(19), CHA3(2)
      WRITE(6,110)
      DO 100 I=1,N
      IF(K(I,1).GT.0) C1=CHA1(K(I,1))
      IF(K(I,1).LE.0) IC1=-K(I,1)
      C2=CHA2(K(I,2))
      C3=CHA3((47-K(I,2))/20)
      IF(K(I,1).GT.0) WRITE(6,120) I, C1, C2, C3, (P(I,J), J=1,5)
  100 IF(K(I,1).LE.0) WRITE(6,130) I, IC1, C2, C3, (P(I,J), J=1,5)
      RETURN
  110 FORMAT(////T11,'I',T17,'ORI',T24,'PART',T32,'STAB',
     &T44,'PX',T56,'PY',T68,'PZ',T80,'E',T92,'M'/)
  120 FORMAT(10X,I2,4X,A2,1X,2(4X,A4),5(4X,F8.1))
  130 FORMAT(10X,I2,4X,1X,I2,2(4X,A4),5(4X,F8.1))
      END
```

```
      BLOCK DATA
      COMMON /PAR/ PUD, PS1, SIGMA, CX2, EBEG, WFIN, IFLBEG
      COMMON /EDPAR/ ITHROW, PZMIN, PMIN, THETA, PHI, BETA(3)
      COMMON /DATA1/ MESO(9,2), CMIX(6,2), PMAS(19)
      COMMON /DATA2/ IDCO(12), CBR(29), KDP(29,3)
      COMMON /DATA3/ CHA1(9), CHA2(19), CHA3(2)
      DATA PUD/0.4/, PS1/0.5/, SIGMA/350./, CX2/0.77/,
     &EBEG/10000./, WFIN/100./, IFLBEG/1/
      DATA ITHROW/1/, PZMIN/0./, PMIN/0./, THETA,PHI,BETA/5*0./
      DATA MESO/7,1,3,2,8,5,4,6,9,7,2,4,1,8,6,3,5,9/
      DATA CMIX/2*0.5,1.,2*0.5,1.,2*0.25,0.5,2*0.,1./
      DATA PMAS/0.,2*139.6,2*493.7,2*497.7,135.,548.8,957.6,
     &2*765.9,2*892.2,2*896.3,770.2,782.6,1019.6/
      DATA IDCO/0,1,6,11,12,13,15,17,19,21,22,25/
      DATA CBR/1.,0.381,0.681,0.918,0.969,1.,0.426,0.662,0.959,
     &0.980,1.,1.,1.,0.667,1.,0.667,1.,0.667,1.,0.667,1.,1.,
     &0.899,0.987,1.,0.486,0.837,0.984,1./
      DATA KDP/1,1,8,2,1,1,2,8,1,1,1,2,3,6,4,7,5,4,6,5,7,2,2,
     &1,2,4,6,2,1,1,1,8,3,2,1,3,8,17,18,1,8,8,2,8,3,8,3,8,2,8,
     &3,3,8,3,5,7,3,9,0,0,8,8,3,8,9,9,14*0,8,4*0,8,0/
      DATA CHA1/'UD','DU','US','SU','DS','SD','UU','DD','SS'/
      DATA CHA2/'GAMM','PI+','PI-','K+','K-','KO','KBO','PIO','ETA',
     &'ETAP','RHO+','RHO-','K*+','K*-','K*O','KB*O','RHOO','OMEG','PHI'/
      DATA CHA3/'    ','STAB'/
      END
```

Table 1

The meaning of the entries of the K and P matrices in the commonblock JET

Row number I corresponds to the I:th particle produced or, after an EDIT call,the I:th particle retained. Within this row,the meaning of each element is as follows.

Element	Description
K(I,1)	K(I,1) > 0: the particle is a primary meson formed by a quark and antiquark in the string, K(I,1) gives the flavour of this pair, numbered as in Table 5
	K(I,1) = 0: a decay product of unknown origin,the connection having been lost by an EDIT call
	K(I,1) < 0: a product coming from the decay of the particle stored in row number - K(I,1)
K(I,2)	particle type numbered as in Table 6
P(I,1)	p_x , particle momentum in the x direction (in MeV/c)
P(I,2)	p_y , particle momentum in the y direction (in MeV/c)
P(I,3)	p_z , particle momentum in the z direction (in MeV/c)
P(I,4)	E , particle energy (in MeV)
P(I,5)	m , particle rest mass (in MeV/c^2), values given in Table 6

Table 2

Default value, allowed range and meaning for the parameters stored in the commonblock PAR and used in JETGEN

Parameter	Default value (Allowed range)	Description
PUD	0.4 ($\frac{1}{3}$ - $\frac{1}{2}$)	probability that a pair materialized in the string will be $u\bar{u}$, the same for $d\bar{d}$, 1-2 • PUD for $s\bar{s}$
PS1	0.5 (0.-1.)	probability that a primary meson will have spin 1, 1-PS1 probability for spin 0
SIGMA (σ)	350. ($\geq$0.)	primary meson $p_\perp$ probability distribution is given by Gaussian $\frac{1}{2\pi\sigma^2} e^{-(p_\perp^2/2\sigma^2)} d^2p_\perp$ (σ in MeV/c)
CX2	0.77 (0.-1.)	probability distribution for $x = \frac{(E+p_z)\text{primary meson}}{(E+p_z)\text{available}}$, 0<x<1 , is given by $(1 - CX2 + 3 \cdot CX2 \cdot (1 - x)^2)\, dx$
EBEG	10000. (>0.)	starting value (in MeV) for the energy available in the generation process
WFIN	100. (>0.)	value (in MeV) for the $E + p_z$ available, below which the generation of primary mesons is stopped
IFLBEG	1 (1,2,3,-1,-2,-3)	flavour of the quark giving rise to the jet, numbered as in Table 4

Table 3

Default value, allowed range and meaning for the parameters stored in the commonblock EDPAR and used in EDIT.

Parameter	Default value (Allowed range)	Description of action of EDIT
ITHROW	1 (0,1,2,3)	ITHROW = 0: keeps all types of particles, stable as well as unstable ones ITHROW = 1: keeps the stable particles, i.e. γ, $\pi^{\pm}$, $K^{\pm}$, K^{o}, and $\bar{K}^{o}$ ITHROW = 2: keeps only γ, $\pi^{\pm}$, and $K^{\pm}$ ITHROW = 3: keeps only $\pi^{\pm}$ and $K^{\pm}$
PZMIN	0.	keeps only the particles with $p_z \geq$ PZMIN
PMIN	0. ($\geq$0.)	keeps only the particles with $p \geq$ PMIN
THETA (θ) PHI (φ)	0. (0.-π) 0. (0.-2π)	rotates the jet axis from the z direction (θ=0) to a given polar angle θ,φ
BETA(1) BETA(2) BETA(3) ($\vec{\beta}$)	0. 0. 0. ($\vec{\beta}^2$<1.)	gives the jet a Lorentz boost, with magnitude and direction determined by $\vec{\beta} = \frac{\vec{v}}{c}$

Table 4

Quark flavours. The positive alternatives are the ones used internally in JETGEN . For IFLBEG the negative values are also available.

IFLBEG (IFL1,IFL2)	Quark
1	u
2	d
3	s
-1	$\bar{u}$
-2	$\bar{d}$
-3	$\bar{s}$

Table 5

Meson flavours, used for K(I,1) > 0 . The flavour labels printed by LIST (column ORI). Flavour mixing between $u\bar{u}$, $d\bar{d}$ and $s\bar{s}$.

K(I,1)	Meson flavour	CHA1	Pseudoscalar mesons	Vector mesons
1	$u\bar{d}$	UD	π^+	ρ^+
2	$d\bar{u}$	DU	π^-	ρ^-
3	$u\bar{s}$	US	K^+	K^{*+}
4	$s\bar{u}$	SU	K^-	K^{*-}
5	$d\bar{s}$	DS	K^o	K^{*o}
6	$s\bar{d}$	SD	$\bar{K}^o$	$\bar{K}^{*o}$
7	$u\bar{u}$	UU	50% π^o , 25% η , 25% η'	50% ρ^o , 50% ω
8	$d\bar{d}$	DD	50% π^o , 25% η , 25% η'	50% ρ^o , 50% ω
9	$s\bar{s}$	SS	50% η , 50% η'	ϕ

Table 6

Particle type codes and masses, used for K(I,2) and PMAS . The particle labels printed by LIST (column PART).

K(I,2)	Particle type	CHA2	Quark composition	PMAS (MeV/c^2)
1	γ	GAMM		0.0
2	π^+	PI+	$u\bar{d}$	139.6
3	π^-	PI-	$d\bar{u}$	139.6
4	K^+	K+	$u\bar{s}$	493.7
5	K^-	K-	$s\bar{u}$	493.7
6	K^0	K0	$d\bar{s}$	497.7
7	$\bar{K}^0$	KB0	$s\bar{d}$	497.7
8	π^0	PI0	$\frac{1}{\sqrt{2}}(u\bar{u}-d\bar{d})$	135.0
9	η	ETA	$\frac{1}{2}(u\bar{u}+d\bar{d})-\frac{1}{\sqrt{2}}\,s\bar{s}$	548.8
10	η'	ETAP	$\frac{1}{2}(u\bar{u}+d\bar{d})+\frac{1}{\sqrt{2}}\,s\bar{s}$	957.6
11	ρ^+	RHO+	$u\bar{d}$	765.9
12	ρ^-	RHO-	$d\bar{u}$	765.9
13	K^{*+}	K*+	$u\bar{s}$	892.2
14	K^{*-}	K*-	$s\bar{u}$	892.2
15	K^{*0}	K*0	$d\bar{s}$	896.3
16	$\bar{K}^{*0}$	KB*0	$s\bar{d}$	896.3
17	ρ^0	RHO0	$\frac{1}{\sqrt{2}}(u\bar{u}-d\bar{d})$	770.2
18	ω	OMEG	$\frac{1}{\sqrt{2}}(u\bar{u}+d\bar{d})$	782.6
19	ϕ	PHI	$s\bar{s}$	1019.6

Table 7

Decay channels and branching ratios, used for IDCO , KDP and CBR in the commonblock DATA2 .

Channel IDC	Decaying particle K(I,2)		Decay products,in KDP(IDC,J),J=1,2,3			Branching ratios (%)	CBR
1	8	π^0	γ	γ		98.8	1.000
2	9	η	γ	γ		38.0	0.381
3			π^0	π^0	π^0	29.9	0.681
4			π^+	π^-	π^0	23.6	0.918
5			γ	π^+	π^-	4.9	0.969
6			γ	γ	π^0	3.1	1.000
7	10	η'	π^+	π^-	η	42.6	0.426
8			π^0	π^0	η	23.6	0.662
9			γ	ρ^0		29.8	0.959
10			γ	ω		2.1	0.980
11			γ	γ		2.0	1.000
12	11	ρ^+	π^+	π^0		100.0	1.000
13	12	ρ^-	π^-	π^0		100.0	1.000
14	13	K^{*+}	K^0	π^+		66.7	0.667
15			K^+	π^0		33.3	1.000
16	14	K^{*-}	$\bar{K}^0$	π^-		66.7	0.667
17			K^-	π^0		33.3	1.000
18	15	K^{*0}	K^+	π^-		66.7	0.667
19			K^0	π^0		33.3	1.000
20	16	$\bar{K}^{*0}$	K^-	π^+		66.7	0.667
21			$\bar{K}^0$	π^0		33.3	1.000

continued

22	17	ρ^0	π^+	π^-		100.0	1.000
23	18	ω	π^+	π^-	π^0	89.9	0.899
24			γ	π^0		8.8	0.987
25			π^+	π^-		1.3	1.000
26	19	ϕ	K^+	K^-		48.6	0.486
27			K^0	$\bar{K}^0$		35.1	0.837
28			π^+	π^-	π^0	14.7	0.984
29			γ	η		1.6	1.000